

Algorithmes : solution du TP

O. Marguin - 4 mai 2011

Exercice 1 : sous-vecteur minimal

```
> restart:
> svm:=proc(v) # algorithme optimal
  local n,somme,i,j,s,g,d;
  n:=nops(v);
  somme:=v[1];g:=1;d:=1;
  s:=v[1];i:=1;
  for j from 2 to n do
    if s>=0 then s:=v[j];i:=j else s:=s+v[j] end if;
    if s<somme then somme:=s;g:=i;d:=j end if
  end do;
  [g,d,somme]
end proc:
> v:=[seq(rand(-10..10)(),i=1..15)];
      v:= [4, 2, -5, -9, 1, -7, -5, -6, 6, -3, -6, -9, 7, 9, 0]
> svm(v);
      [3, 12, -43]
>
```

Exercice 2 : puissance rapide

```
> restart:
> puiss:=proc(x,n)
  if n=0 then 1
  elif type(n,even) then puiss(x*x,n/2)
  else x*puiss(x,n-1)
  end if
end proc:
> puiss(3,7)-3^7;
      0
```

Exercice 3 : calcul des coefficients du binôme

```
> restart:
> C:=proc(n,p)
  if p=0 or p=n then 1 else C(n-1,p-1)+C(n-1,p) end if
end proc:
> C(7,3);
      35
> C(40,20); # on a beau attendre, aucun résultat ne sort !
Warning, computation interrupted
```

La complexité en espace est proportionnelle à la profondeur de l'arbre binaire exploré, qui est inférieure ou égale à n .

Un dénombrement facile montre que le nombre d'appels à la fonction C provoqués par l'instruction $C(n,p)$ est égal à $2C(n,p) - 2$, il s'ensuit que le nombre d'additions est égal à $2C(n-1,p-1) - 2$. Le pire des cas est pour $p =$ partie entière de $n/2$ et d'après la formule de Stirling, la complexité en temps est un grand theta de $\exp(n)$: l'algorithme est inefficace.

On évite les appels redondants grâce à l'option `remember`, les résultats intermédiaires sont alors stockés dans une table :

```
> Cbis:=proc(n,p)
```

```
option remember;
  if p=0 or p=n then 1 else Cbis(n-1,p-1)+Cbis(n-1,p) end
if
end proc:
> Cbis(40,20);
```

137846528820

Cette fois les complexités en espace et en temps sont au pire en n^2 .

L'algorithme optimal (linéaire) utilise la récurrence : $C(n,p) = n/p * C(n-1,p-1)$, ce qui donne :

```
> Cter:=proc(n,p)
  if p=0 then 1 else n/p*Cter(n-1,p-1) end if
end proc:
> Cter(100,50);
```

100891344545564193334812497256

```
>
```